
POLARIZATION ANALYSIS

In this technical guide, the techniques and commands used to perform polarization 
analysis in ASAP®(Advanced Systems Analysis Program) from Breault Research 
Organization, Inc., are described. In ASAP, you can include the effects of 
polarization in your analysis. This polarization analysis can be performed in either 
of the ASAP modes, BEAMS INCOHERENT GEOMETRIC or BEAMS COHERENT 
DIFFRACT.

This technical guide is structured in the same order that is commonly used to 
perform any ASAP analysis:

• Setting up the system: system commands, coatings, polarization components

• Creating polarized sources

• Ray tracing polarized light

• Analyzing polarized light

The content of this document was written with the assumption that the reader has 
an understanding of polarized light. For those interested in the fundamental 
concepts of polarized light, please see “List of References” on page 31. 

Specific ASAP commands are capitalized when referred to inside this document. 
Examples of ASAP scripts presented in this document were created in the ASAP 
Editor. The Editor document window is shown once in the first example. 
Subsequent examples show only the script.

This technical guide is complemented by the technical publication bro4310, 
“Polarization: Improved simulation of real-world devices in ASAP”, which 
describes polarization features added in ASAP 2008 and later. This and any other 
BRO technical publication referenced in this document may be viewed or 
downloaded from the BRO Knowledge Base, http://www.breault.com/k-base.php.

P O L A R I Z A T I O N  A N A L Y S I S

Polarization Commands
. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .P O L A R I Z A T I O N  C O M M A N D S

Most of the commands used in polarization analysis are the same as those used in 
scalar analysis. This document discusses performing polarization analysis in 
ASAP using 12 commands. Some of these 12 commands are used for only 
polarization analysis, while others are also used in scalar analysis; but, to some 
extent, they are used differently in polarization analysis. While there are some 
more advanced or specialized ASAP polarization techniques not discussed within 
this document, the techniques covered here allow you to handle the vast majority 
of polarization issues.

The table below lists the 12 commands organized into five command types.

TABLE 1. ASAP Commands for Polarization Analysis

1The POLARIZ command is used as both a source creation and an analysis 
command.

. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .S E T T I N G  U P  T H E  S Y S T E M

Ini t ia l  system commands
We must issue two general commands to perform ASAP analysis with polarized 
light. These commands are FRESNEL BOTH and XMEMORY (either NORM or FULL). 
If you are interested in modeling birefringent crystals or total internal reflections, 
you must also issue the SPLIT command.

Command Type Command Name

System FRESNEL BOTH
XMEMORY

Coating INTERFACE
COATING LAYERS
COATING MODELS

Polarization component IDEAL
CRYSTAL

Source creation POLARIZ1

Analysis PLOT POLARIZATION
LIST ELLIPSE
FIELD
FIELD...DELTA
POLARIZ1
 8 ASAP Technical Guide



. .
 . 

. .P O L A R I Z A T I O N  A N A L Y S I S

Setting Up the System
The FRESNEL BOTH command instructs ASAP to apply the Fresnel equations at 
media boundaries to obtain the appropriate amounts of reflected and transmitted 
light. BOTH refers to calculating the S and P polarization components separately for 
both reflection and polarization, leading to two reflection and two transmission 
coefficients. In the default form, FRESNEL AVE, the S and P polarization 
components are averaged together, thus leaving only one reflection and one 
transmission coefficient. The application of polarization-sensitive coatings to 
media boundaries is discussed in “Polarization-sensitive coatings” on page 9.

The XMEMORY command determines the amount of information that accompanies 
each ray. It is always set at one of three levels: MIN, NORM, or FULL. XMEMORY MIN 
stores the least amount of ray data. To perform polarization analysis in the BEAMS 
INCOHERENT GEOMETRIC mode, we must issue the XMEMORY NORM command, 
which is the default setting. This command instructs ASAP to include additional 
information with each ray that describes its polarization state. It is not necessary to 
explicitly issue an XMEMORY command when working in the BEAMS COHERENT 
DIFFRACT mode, because issuing the PARABASAL command automatically sets 
the XMEMORY to FULL.

Polar izat ion-sensi t ive  coat ings
Three ways exist to assign a polarization-sensitive coating to an OBJECT in ASAP. 
The simplest method is to use the INTERFACE command and choose numerical 
reflection and transmission values for normal incidence. ASAP determines a 
quarter-wave layer that achieves these values at normal incidence. At all other 
angles of incidence, ASAP calculates the correct Fresnel reflection and 
transmission coefficients for the incident polarization and angle of incidence based 
on the quarter-wave layer it determined. Although this method is not based on the 
users actual system, it is simple and at times may be adequate.

The most precise method for specifying a polarization-sensitive coating in ASAP 
is with the COATING LAYERS command. Two basic ways to use the COATING 
LAYERS command exist. You can enter the layer thicknesses as either the physical 
thickness in WAVELENGTH units, or as the optical thickness (the product of the 
index and the physical thickness) in units of the reference vacuum wavelength. 
The syntax for COATING LAYERS is:

COATING LAYERS [W]

d  m  [ d’  m’  d’’  m’’ ... ] [‘name’]
ASAP Technical Guide  9

Setting Up the System
NOTES: 

(1) If W is entered, the layer thicknesses are optical thicknesses in 
units of the reference wavelength in a vacuum. If W is absent, the 
layer thicknesses are physical thicknesses specified in 
wavelength units.

(2) Layers are defined starting from the low-index side of the 
interface and ending on the high-index side of the interface. This 
order is usually the opposite of the order in which the real layers 
are applied to a substrate.

(3) Groups of d layers can be replicated m times by placing a -d and 
a positive m after a group of layers.

Two examples of a COATING LAYERS input, which were created in the ASAP 
Editor, are shown below. The first is a simple quarter-wave magnesium fluoride 
(MGF) coating, using the form with a reference wavelength (W) specified. The 
second example describes physical layer thicknesses in wavelength units, and uses 
the layer group replication method described in Note (3) above.

The first example shows a script with the COATING LAYERS command, and 
specifies the layer as fractional optical waves of the reference wavelength (550 nm 
in this case). In other words, the product of the physical thickness of the layer 
times the MGF index (1.38) is equal to 0.25  550 nm.

W reference wavelength in vacuum

d layer thickness—either (1) physical thickness 
in wavelength units, or (2) optical waves of 
reference wavelength

m names (or numbers) of previously defined 
media
 10 ASAP Technical Guide



. .
 . 

. .P O L A R I Z A T I O N  A N A L Y S I S

Setting Up the System
Example 1 of COATING LAYERS command (shown in the ASAP Editor document window)

The second example of script with the COATING LAYERS command specifies the 
layer as physical thicknesses in WAVELENGTH units (NM, in this case). It uses the 
group replication feature to replicate the group of two layers (MGF and FICT) five 
times.

Example 2 of script for COATING LAYERS command
ASAP Technical Guide  11

Setting Up the System
The final method for defining a polarization-sensitive coating uses the COATING 
MODELS command. With this command, you can assign specific reflection and 
transmission values to an OBJECT as a function of incident angle and polarization. 
These values are supplied in the form of a look-up table, which is called by the 
BSDFDATA command and the ...ANGLES command argument. This command is 
often useful if the exact coating prescription is not known, but measured reflection 
and transmission values are available. It can also be useful to model more realistic 
coating behavior than the ideal prescription due to errors in layer thicknesses and 
indices.

COATING MODELS has two syntax forms. These two forms are similar, except that 
the more complex form allows for the use of different models for different 
wavelengths. This document addresses the simpler form. The slightly different 
syntax for the other form can be found in either ASAP online Help or the ASAP 
Reference Manual (see the Command Output window for a link to the manual, or 
look in <install dir>\asapxx\doc\manual.txt or manual.pdf). 

Setting Up the System
The syntax for the simpler (without wavelength dependence) COATING MODELS 
command is given below.

Example of script for COATING MODELS command

The numbers shown as 1 2 3 4 in the example above refer to the MODELS 
corresponding to the S-polarization reflection, P-polarization reflection, S-
polarization transmission, and P-polarization transmission, respectively. We can 
refer to the same MODEL for more than one of the four categories, such as 

COATING MODELS 1 1 2 2 

or even,

COATING MODELS 1 1 1 1 

The two numbers on the next line of the command give the normal incidence 
reflectance and transmittance coefficients. These numbers are followed by the 
name assigned to the coating. 

The form of the entry for the BSDFDATA table is shown below. 

Example of a BSDFDATA table, to which the COATING MODELS command refers

The first column of numbers on the left of the BSDFDATA table lists angle of 
incidence values. For use in COATING MODELS, the entries in the middle column 
of numbers are always 0’s. The last column of numbers on the right contains the 
flux values. Once constructed, this MODEL can be referred to in any of the four 

Model numbers

Name of coating

Normal incidence transmission

Normal incidence reflectivity

Model numbers

Name of coating

Model numbers

Name of coating

Normal incidence transmission

Normal incidence reflectivity
ASAP Technical Guide  13



P O L A R I Z A T I O N  A N A L Y S I S

Setting Up the System
positions in the COATING MODELS command. The given flux values represent only 
one of the two components of the actual ray reflection and transmission values. 
The reflection and transmission amounts for a ray at a given angle are obtained by 
multiplying the flux values from the BSDFDATA table by the appropriate (r or t) 
normal incidence value given in the second line of the COATING MODELS 
command.

NOTE: Several peculiarities exist with this command, which are due 
to its original purpose for defining scatter models. The zero 
degree angle of incidence value is always normalized to a value of 
one. The other values are then scaled by the same factor. Values 
for angles intermediate to those listed explicitly in the table are 
interpolated linearly in logarithmic space. Therefore, to get more 
accurate interpolation for intermediate values, you need to 
explicitly specify many table values at close increments, rather 
than having large gaps between angle entries. Also, because 
logarithms are involved, you need to specify zero flux values with 
very small numbers, such as 1E-9, rather than exactly zero.
 14 ASAP Technical Guide

Setting Up the System
An example of ASAP script that defines MODELS and COATING MODELS is shown 
below.

Example of MODELS and COATING MODELS, and applying COATING MODEL to an 
OBJECT

Name of 
interface 
coating for 
hypotenuse

Name of 
interface 
coating for 
hypotenuse
ASAP Technical Guide  15

Setting Up the System
Polar izat ion components
Two commands exist in ASAP for describing polarization components: IDEAL (a 
LENS type) and CRYSTAL (a MEDIA type). The IDEAL command describes the 
polarization properties of the component in the form of a Jones operator (a 22 
matrix). The CRYSTAL command describes a uniaxial birefringent medium. These 
commands are described in more detail below.

I D E A L

The IDEAL command is a type of LENS entity. For polarizing elements, the values 
in the row following the IDEAL command are always set to 1 0 0 1. The next 
row is the Jones operator in a row format rather than the conventional 22 format.
 16 ASAP Technical Guide

Setting Up the System
TABLE 2. Examples of equivalent Jones operators in their conventional 2x2 format, and in the ASAP 
four-value row format 

Polarizing Element Jones Matrix ASAP Representation

Vertical linear polarizer 0 0 0 1

Horizontal linear polarizer 1 0 0 0

45° linear polarizer 0.5 0.5 0.5 0.5

-45° linear polarizer 0.5 -0.5 -0.5 0.5

Quarter-wave plate
(fast axis vertical)

COS[45] ‘SIN[45] 0 0 SIN[45] ‘-COS[45]

Quarter-wave plate
(fast axis horizontal)

COS[45] ‘SIN[45] 0 0 -SIN[45] ‘COS[45]

Half-wave plate
(fast axis vertical or 
horizontal)

1 0 0 -1

Half-wave plate
(fast axis ±45°)

0 1 1 0
ASAP Technical Guide  17

Setting Up the System
Two ways to generate the same polarizing element with the IDEAL command

Right circular polarizer 0.5 0‘-0.5 0‘-0.5 0.5

Left circular polarizer 0.5 0‘-0.5 0‘0.5 0.5

Polarizing Element Jones Matrix ASAP Representation

"a b c d" matrix elements

Horizontal polarizer
Jones matrix elements

Rotate polarizer 45°

Case 1

Case 2

45° polarizer
Jones matrix elements

"a b c d" matrix elements

Horizontal polarizer
Jones matrix elements

Rotate polarizer 45°

Case 1

"a b c d" matrix elements

Horizontal polarizer
Jones matrix elements

Rotate polarizer 45°

Case 1

Case 2

45° polarizer
Jones matrix elements

Case 2

45° polarizer
Jones matrix elements
 18 ASAP Technical Guide

Setting Up the System
C R Y S T A L

One of the media supported in ASAP is crystal. The MEDIA CRYSTAL command 
establishes a uniaxial birefringent medium. Within the command, we must specify 
the following information:

• Index of refraction for rays polarized along the crystal (or optic) axis.

• X, Y, and Z component values for any vector-oriented parallel to the crystal axis 
(or, equivalently, the direction cosines).

• Medium that describes the ordinary index (polarized perpendicular to the 
crystal axis). This entry references a medium that had to be defined prior to the 
MEDIA CRYSTAL command.

• Name of the crystal.

Example of a birefringement crystal constructed with the MEDIA CRYSTAL command

Note that the incident ray is split into an ordinary ray and an extraordinary ray. The 
ordinary ray is considered the parent ray in ASAP, while the extraordinary ray is 
the child ray. In general, the SPLIT command must be set high enough to allow for 
one increase in generation for every time the ray propagates through the crystal. 
Issues related to calculating the correct field energy and polarization state for 
beams after propagation through crystals are addressed in the section, “Analyzing 
Polarized Light” on page 24.

BRO recommends defining a separate CRYSTAL for each component in a system. 
The reason is that, if an object that references a CRYSTAL is moved, the 
coordinate transformation is applied to the medium and therefore changes all 
OBJECTS that reference that medium.
ASAP Technical Guide  19

Creating Polarized Sources
. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .C R E A T I N G  P O L A R I Z E D  S O U R C E S

The POLARIZ command must be issued prior to polarized source creation. This 
command sets the polarization for all future source creation. The polarization state 
given in the POLARIZ command stays in effect until a different POLARIZ 
command is issued. The POLARIZ command is also used to examine the properties 
of the individual X, Y, or Z field components. This second type of application of 
the POLARIZ command is described later in this guide in the analysis section.

Similar to the common Jones vector representation, the POLARIZ command 
describes the polarization state of the light by specifying the complex amplitude of 
two orthogonal polarization states. The ASAP syntax for creating the intended 
state of polarization is shown below.

Example of a POLARIZ source

The first entry after the word POLARIZ is one of the global X, Y, or Z axes. The 
value of the a entry specifies the complex amplitude of the polarization component 
in this specified global axis direction. The a’ term that follows the a term gives 
the complex amplitude of the polarization component orthogonal to both the a 
component and the ray direction. ROTATE or SOURCE DIRECTION is applied after 
ray creation, and gives rise to polarization directions that remain orthogonal to the 
ray propagation direction. Examples using the POLARIZ command to specify 
various polarization states are given below.
 20 ASAP Technical Guide

Creating Polarized Sources
Examples of POLARIZ command to specify four polarization states

You can specify a linear polarization state that is polarized along any of the three 
global axes without explicitly specifying a and a’ values. An example of this is 
given below.

POLARIZ X

Several other ways exist for specifying the same polarization state. For example, 
the three POLARIZ commands shown below all specify the same X-polarization 
state.

POLARIZ X

POLARIZ X 1 0

POLARIZ Y 0 1

The ASAP source-creation commands, GRID, GAUSSIAN, RAYSET, and 
DECOMPOSE, create totally polarized light. As stated previously, the polarization 
state is input through the POLARIZ command in a form similar to Jones vectors. 
States of partial polarization can be created only through the incoherent summing 

Horizontal linear

Vertical linear

Left-hand circular

Right-hand Circular

Horizontal linear

Vertical linear

Left-hand circular

Right-hand Circular
ASAP Technical Guide  21



P O L A R I Z A T I O N  A N A L Y S I S

Creating Polarized Sources
of combinations of differently polarized sources. In ASAP, it is not possible to 
input polarization states directly in the form of Stokes vectors. When a POLARIZ 
command is issued prior to an EMITTING command, the chosen polarization state 
is applied to all the rays prior to the randomizing of their directions. Next, the 
directions of the rays get randomly perturbed. Because of this, the user should be 
careful when using POLARIZ in conjunction with EMITTING. 

Two examples of sources generated by EMITTING and POLARIZ and are shown 
below. Each example shows the script and the plot output.

Example #1 of script using EMITTING and POLARIZ linear horizontal

Example #1 of plot generated from EMITTING and POLARIZ linear horizontal
 22 ASAP Technical Guide

Creating Polarized Sources
Example #2 of script using EMITTING and POLARIZ left-hand circular

Example #2 of plot generated from EMITTING and POLARIZ left-hand circular

The DECOMPOSE command creates a set of beams that correspond to the field 
within the current BRO029.dat file. It is one of the four commands, along with 
GRID, RAYSET, and GAUSSIAN, that allow for the creation of COHERENT sources. 
As such, if polarization analysis is being performed, a POLARIZ command must be 
issued prior to the DECOMPOSE command. Also, if the field to be decomposed has 
significant amounts of orthogonal polarization components, a POLARIZ and 
DECOMPOSE pair must be issued for each component. 
ASAP Technical Guide  23

Ray Tracing Polarized Light
An example is shown below.

POLARIZ X

DECOMPOSE POSITION

POLARIZ Y

DECOMPOSE POSITION

POLARIZ Z

DECOMPOSE POSITION

NOTE: Because of the method employed in ASAP, the results are 
sometimes more accurate when you skip one of the polarization 
components if the amount of energy in that component is several 
orders of magnitude less than the other components. In other 
words, if the field is polarized primarily in X and Y with only a tiny 
Z component, it may be more accurate to just decompose the X 
and Y components and skip the Z decomposition. To determine 
whether the component with the small amount of energy should 
be decomposed or not, we must examine how well the total fields 
match before and after DECOMPOSE. 

. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .R A Y  T R A C I N G  P O L A R I Z E D  L I G H T

When tracing polarized light, ASAP carries additional information for each ray. 
Even so, ray tracing with the TRACE command is exactly the same for polarized 
light as it is for scalar light, provided that the commands described in the previous 
two sections that relate to setting up the system and creating polarized sources 
have been done correctly.

. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .A N A L Y Z I N G  P O L A R I Z E D  L I G H T

Some of the commands used to analyze polarized light are most appropriate for 
use in the BEAMS INCOHERENT GEOMETRIC mode. These commands can still be 
used in the BEAMS COHERENT DIFFRACT mode, but portions of their output 
should be ignored. Also, some commands used to analyze polarized light can be 
used only in the BEAMS COHERENT DIFFRACT mode. These distinctions are 
addressed below.

Analyzing Polarized Light
Polar izat ion analysis in  the BEAMS INCOHERENT GEOMETRIC mode
In ASAP, all the incoherent flux commands work the same in the analysis of 
polarized light as they do for analysis without polarization. Two additional 
commands are available for polarization analysis in the BEAMS INCOHERENT 
GEOMETRIC mode. Those commands are PLOT POLARIZATION and LIST 
ELLIPSE.

P L O T  P O L A R I Z A T I O N

PLOT POLARIZATION (also referred to in its simpler form as PLOT POLAR) plots 
the polarization ellipses of all the CONSIDERed and SELECTed rays that are inside 
the current WINDOW on a ray-by-ray basis. The size of the individual ellipse is 
related to the amount of ray flux. Arrows indicate the handedness of the 
polarization state. When used in conjunction with a MEDIA CRYSTAL command, it 
plots the ordinary and extraordinary rays separately, without information as to their 
relative phases. For this reason, we must use the FIELD...DELTA command to get 
the correct polarization state after propagation through MEDIA CRYSTAL. The 
FIELD command can be used only in BEAMS COHERENT DIFFRACT mode, and is 
discussed in “Polarization analysis in the BEAMS COHERENT DIFFRACT 
mode” on page 28. 
ASAP Technical Guide  25

Analyzing Polarized Light
A script example of the PLOT POLAR command is shown below.

Examples of the PLOT POLAR command with PLOT RAYS and corresponding 
polarization plots

L I S T  E L L I P S E

LIST ELLIPSE lists the polarization parameters of all the CONSIDERed and 
SELECTed rays on a ray-by-ray basis in the ASAP command output window. The 
three parameters listed are:

1 The orientation is described by listing the components of a unit vector (direction 

cosines) in the direction of the major axis of the polarization ellipse.

Analyzing Polarized Light
2 The ellipticity is described as the ratio of the length of the minor axis divided by 

the length of the major axis. This has a value of zero for linearly polarized light, 

one for circularly polarized light, and a value between zero and one for all other 

polarization states. 

3 The handedness is described by the sign of the ellipticity. If the sign is positive, 

the polarization state is right-handed; if it is negative, the polarization state is left-

handed. Right-handed is defined as the polarization direction rotating clockwise 

in time for light approaching the observer.

Similar to the PLOT POLAR command, the LIST ELLIPSE command lists the 
ordinary and extraordinary rays separately, without any relative phase information. 
For this reason, it is necessary to use the FIELD...DELTA command to get the 
correct polarization state after propagation through MEDIA CRYSTAL.

Example of the LIST ELLIPSE command
ASAP Technical Guide  27

Analyzing Polarized Light
Polar izat ion analysis in  the BEAMS COHERENT DIFFRACT mode

F I E L D  A N D  T H E  D E L T A  O P T I O N

The FIELD command is used in the COHERENT mode to calculate the complex 
field, regardless of whether we are examining a scalar field or a vector field 
including polarization. In the scalar case, ASAP calculates one complex field 
value per pixel; whereas, in the polarized case, it calculates three complex field 
values per pixel, one for each of the global X, Y, Z polarization components. 
Because of this, the maximum number of pixels available in COHERENT 
polarization analysis goes down by the square root of three in each dimension. The 
SPREAD NORMAL command can be used to calculate the field energy for 
COHERENT scalar fields, but it does not account for polarization, and therefore 
gives incorrect results when used for polarized fields. Use the FIELD ENERGY 
command to calculate the energy for polarized fields. This command correctly 
sums the squared moduli of the X, Y, and Z components to obtain the total field 
energy.

To examine the various field components of the field for a given polarization in 
ASAP, we must first issue a POLARIZ command, which specifies the component of 
interest. (A different usage of this command was described earlier to set the 
polarization state for future source creation.) This step applies for any of the field 
parameters except FIELD ENERGY, which as stated above, sums all the 
components together in quadrature. The example below shows the commands we 
must use to examine the phase of each component after issuing a FIELD command 
in any form.

POLARIZ X

DISPLAY 29 PHASE  !! displays x-pol phase

POLARIZ Y

DISPLAY 29 PHASE  !! displays y-pol phase

POLARIZ Z

DISPLAY 29 PHASE  !! displays z-pol phase

Analyzing Polarized Light
The POLARIZ command can also be used to determine the amount of energy in 
each component. When used in conjunction with the FORM 2 command to square 
the amplitude (or modulus) values on a pixel-by-pixel basis, we obtain an energy 
map for a given polarization component. 

An example is given below.

POLARIZ X

DISPLAY 29 AMPLITUDE

FORM 2 !! generates array of x-pol energy values

POLARIZ Y

DISPLAY 29 AMPLITUDE

FORM 2 !! generates array of y-pol energy values

POLARIZ Z

DISPLAY 29 AMPLITUDE

FORM 2 !! generates array of z-pol energy values

Analyzing Polarized Light
The FIELD command used with the DELTA option creates a plot of the polarization 
ellipses. Unlike the PLOT POLAR command, the plots created by FIELD...DELTA 
sum the overlapping beam fields with the appropriate relative phases. This process 
allows for the correct plotting of the polarization ellipses, even after the beams 
have split into ordinary and extraordinary beams. An example is shown below.

Example of the same field examined with both the PLOT POLAR command and the FIELD...DELTA command

SCRIPT EXAMPLES FOR POLARIZATION A
The following scripts are examples of polarization in ASAP. You can copy and 
paste the script into an editor, such as the ASAP Editor, and run the file to see the 
results.

Crystal Interference Pattern

Creates a birefringent medium with the CRYSTAL command and polarizer with the 
IDEAL command to create an interference pattern from a diverging beam.

!!ASAP EXAMPLE OF INTERFERENCE FROM A BIREFRINGENT
!!UNIAXIAL CRYSTAL
SYSTEM NEW
RESET

UNITS MM
WAVELENGTH 0.0001

MEDIA
  1.6 'ORD'
  1.5 CRYSTAL 0 0 1 ORD 'CRYSTAL'

FRESNEL BOTH

SURFACE
  PLANE Z 1 RECT 1.5
OBJECT 'SIDE_1'
  INTERFACE COAT +BARE AIR CRYSTAL

SURFACE
  PLANE Z 2 RECT 1.5
OBJECT 'SIDE_2'
  INTERFACE COAT +BARE AIR CRYSTAL

LENS
  IDEAL Z 3 2
    1 0 0 1 define a no lens
    0 0 0 1 turn it into a vertical linear polarizer
OBJECT 'LINEAR_POLARIZER'

SURFACE
ASAP Technical Guide  i



  PLANE Z 4 RECT 2.5
OBJECT 'DETECTOR'

BEAMS COHERENT DIFFRACT
PARABASAL 4
WIDTHS 1.6

POLARIZ X 1 0
GRID POLAR Z 0 0 1 0 360 5 6
SOURCE POSITION 0 0 -5

TITLE 'BIREFRINGENT PLATE WITH LINEAR POLARIZER'

WINDOW Y Z

PROFILES OVERLAY

TRACE PLOT

CONSIDER ONLY DETECTOR
WINDOW Y -2@2 X -2@2
PIXELS 301

FIELD ENERGY 4

DISPLAY
PICTURE

RETURN

Double Refraction Ray Trace

Generates a plot of double refraction for a ray trace through a calcite crystal.

!!ASAP FILE OF ORDINARY AND EXTRAORDINARY RAY
!!SPLITTING IN CALCITE
SYSTEM NEW
RESET

UNITS MM
WAVELENGTH 630 NM

FRESNEL BOTH

MEDIA
  1.655 'CALCITE_ORD'
  1.485 CRYSTAL COS[30] 0 SIN[30] CALCITE_ORD 'CALCITE_EXT'

SURFACE
  PLANE Z 0 RECT 10
OBJECT 'CALCITE_FRONT'
  INTERFACE COATING +BARE AIR CALCITE_EXT

SURFACE
  PLANE Z 10 RECT 10
OBJECT 'CALCITE_BACK'
  INTERFACE COATING +BARE AIR CALCITE_EXT

BEAMS INCOHERENT GEOMETRIC

POLARIZ X 1 1
RAYSET Z -5
0 0 1
SOURCE DIR 0 0 1

WINDOW X Z
TITLE 'ORDINARY AND EXTRAORDINARY RAY TRACES THROUGH CALCITE'
PLOT FACETS OVERLAY
TRACE PLOT

Birefringent Sapphire Dome

Creates sapphire dome with the CRYSTAL command, traces rays through dome, 
and calculates polarization map of field.

!! Polarization Effects of a Sapphire Dome
SYSTEM NEW
RESET

WAVELENGTH 4250 UNITS 1.E-6

MEDIA
  1.664 'ORDSAP' !ordinary indices for following extraordinary
  1.656 CRYSTAL 0 0 1 ORDSAP 'SAPPHIRE'

FRESNEL BOTH

SPLIT 1

SURFACE
  OPTICAL Z 0 37.5 ELLIPSE 37.5
OBJECT 'DOME_OUTER'
  INTERFACE COATING +BARE AIR SAPPHIRE

SURFACE
  OPTICAL Z 0 35.214 ELLIPSE 35.214
OBJECT 'DOME_INNER'
  INTERFACE COATING +BARE AIR SAPPHIRE
  SHIFT Z 2.286

SURFACE
  PLANE Z 0 ELLIPSE 40
OBJECT 'IMAGE'
  SHIFT Z 45

BEAMS INCOHERENT GEOMETRIC

GRID ELLIPTIC Z -1 -1 1 -30 30 1 5
SOURCE DIRECTION Z

PROFILES OVERLAY 'SAPPHIRE DOME & RAY TRACE'
TRACE PLOT

RAYS 0
BEAMS COHERENT DIFFRACT
PARABASAL 8
WIDTHS 1.6

POLARIZ X 1 1'90
GRID ELLIPTIC Z -5 -4@30 2@25
SOURCE DIRECTION Z

WINDOW Y X
ARROWS 1
SEGMENTS 2
PLOT POL 'INPUT POLARIZATION'

TRACE
PLOT POLAR 'OUTPUT POLARIZATION VIA PLOT POLAR'

TITLE 'ACTUAL ELECTRIC FIELD BEHAVIOUR'
PIXELS 15 set pixels for field real points
FIELD REAL 45 DELTA

. .
$DO 1 4
FIELD REAL 45 DELTA +?/8 OVERLAY COLOR ?

RETURN

Crystal Quarter-Wave Plate

Constructs a quarter-wave plate using the CRYSTAL command, traces rays, and 
generates polarization map of beam with both PLOT POLARIZ and 
FIELD...DELTA.

!!ASAP FILE OF A CALCITE QUARTER WAVE PLATE
!!FAST AXIS VERTICAL
SYSTEM NEW
RESET

UNITS MM
FRESNEL BOTH

MEDIA
(NO=1.655) 'CALCITE_ORD'
(NE=1.485) CRYSTAL 0 1 0 CALCITE_ORD 'CALCITE_EXT'

!!FIRST CALCULATE THE REQUIRED THICKNESS OF QUARTER WAVE PLATE
!!PHASE=2PI/LAMBDA*D*(NO-NE)=PI/2 FOR QUARTER WAVE PLATE

LAMBDA=0.6328E-3
D=LAMBDA/(4*ABS(NO-NE))

!!SOME OTHER VARIABLES TO DEFINE THE LASER
W0=0.5 !!LASER WAIST SIZE 1/e^2 RADIUS
PI=4*ATAN(1)

SURFACE
  PLANE Z 0 RECT 10
OBJECT 'Q_WAVE_FRONT'
  INTERFACE COATING +BARE AIR CALCITE_EXT

SURFACE
  PLANE Z (D) RECT 10
OBJECT 'Q_WAVE_BACK'
  INTERFACE COATING +BARE AIR CALCITE_EXT

SURFACE
  PLANE Z 1 RECT 10
OBJECT 'DETECTOR'

!!Macro to create Polarized GRID POLAR

POLAR_RAY { 3
BEAMS COHERENT
WAVELENGTH 632.8 NM
PARABASAL 4
WIDTH 1.6

POLARIZ X #1 #2
USERAPOD POS 2@(W0*SQRT(PI/2)) W0*SQRT(PI/2)^2
GRID POLAR Z -1 0 4.6*W0/2 0 360 11 6
SOURCE DIR SIN[#3] 0 COS[#3]
}
ENTER POLARIZ X a POLARIZATION COMPONENT>
ENTER POLARIZ X a' POLARIZATION COMPONENT>
ENTER INCIDENT ANGLE>

!!Macro to do the polarization analysis
ANALYSIS {
$POLAR_RAY 1 1 0

WINDOW Y -2@(4.6*W0/2) X -2@(4.6*W0/2)
WINDOW 1.25
PIXELS 51

FIELD ENERGY -1
DISPLAY
ISOMETRIC 'INPUT LASER BEAM'
RETURN

PLOT POLAR 'PLOT POLAR ILLUSTRATION OF INPUT'
PIXELS 15

FIELD REAL -1 DELTA 'POLARIZATION ELLIPSES WITH FIELD...DELTA'

$DO 1 4
FIELD REAL -1 DELTA +?/8 OVERLAY 'ELECTRIC FIELD VECTORS WITH FIELD...DELTA!'
RETURN

WINDOW Y Z
TITLE 'RAY TRACE THROUGH CALCITE'
PROFILES OVERLAY
TRACE PLOT 10 TEXT OVERLAY
0 3 -5, 0 0 0.25, 0 0.25 0 'QUARTER WAVE PLATE'
0 2 2, 0 0 0.25, 0 0.25 0 'DETECTOR'
RETURN
RETURN

CONSIDER ONLY DETECTOR
WINDOW Y -2@(4.6*W0/2) X -2@(4.6*W0/2)
WINDOW 1.25
PIXELS 51

FIELD ENERGY 1
DISPLAY
ISOMETRIC 'LASER BEAM AFTER QUARTER WAVE PLATE'
RETURN

PLOT POLAR 'PLOT POLAR ILLUSTRATION OF OUTPUT'
PIXELS 15

FIELD REAL 1 DELTA 'OUTPUT POLARIZATION ELLIPSES WITH FIELD...DELTA'
$DO 1 4
FIELD REAL 1 DELTA +?/8 OVERLAY 'OUTPUT ELECTRIC FIELD VECTORS WITH FIELD...DELTA' COLOR ?
}

$ANALYSIS
RETURN

Dichroic Polarizer

Uses the CRYSTAL command to create dichroic polarizer, traces rays, and 
calculates transmitted polarization.

!!ASAP FILE TO DEMONSTRATE A DICHROIC POLARIZER
SYSTEM NEW
RESET

UNITS MM
FRESNEL BOTH
BEAMS INCOHERENT GEOMETRIC

CUTOFF (C=1E-9)

A=-LOG(C)/10 !!ABSORPTION COEFFCIENT
MEDIA
  1.6 ABSORB (A) 'ORDINARY'
  1.5 CRYSTAL COS[30] 0 SIN[30] ORDINARY 'DICHROIC'

SURFACE
  PLANE Z 0 RECT 10

OBJECT 'FRONT'
  INTERFACE COATING +BARE AIR DICHROIC

SURFACE
  PLANE Z 10 RECT 10
OBJECT 'BACK'
  INTERFACE COATING +BARE AIR DICHROIC

POLARIZ X 1 1
RAYSET Z -5
0 0 1
SOURCE DIRECTION 0 0 1

WINDOW Y -2@2 X -2@2
PLOT POLAR 'POLARIZATION STATE BEFORE DICHROIC POLARIZER'
LIST ELLIPSE

TITLE 'DICHROIC POLARIZER: EXTINGUISHING THE EXTRAORDINARY RAY'
WINDOW X Z
PLOT FACETS OVERLAY
TRACE PLOT

WINDOW Y -2@2 X -2@2
PLOT POLAR 'POLARIZATION STATE AFTER DICHROIC POLARIZER'
LIST ELLIPSE

Glan Prism Polarizer

Constructs a Glan polarizing prism using the CRYSTAL command, traces ray 
through polarizer, and examines the polarization states of the beams emerging 
from the top and back of the prism. The polarization of the top beam, which should 
be a mixed polarization state, is simplified by assigning a +BARE INTERFACE on 
the diagonal surfaces of the prism. Therefore, the reflected beam through the prism 
top is only the TIR beam of the single polarization component.

!!ASAP FILE OF GLAN FOUCAULT PRISM
SYSTEM NEW
RESET

UNITS MM
WAVELENGTH 630 NM
FRESNEL BOTH
SPLIT 2
BEAMS INCOHERENT GEOMETRIC

MISSED ARROWS 5

MEDIA
  1.655 'CALCITE_ORD'
  1.485 CRYSTAL 1 0 0 CALCITE_ORD 'CALCITE_EXT'

AIR_GAP=0.25

SURFACE
  PLANE Z 0 RECT  5 ((5-(AIR_GAP/2))/TAN[38.5])
OBJECT 'CALCITE_FRONT'
  INTERFACE COATING +BARE AIR CALCITE_EXT

SURFACE
  PLANE Y (5-(AIR_GAP/2))/TAN[38.5]
  PLANE Z 0
  PLANE Z 5-(AIR_GAP/2) RECT 5 25
  ROTATE X 38.5 0 5-(AIR_GAP/2)
OBJECT 'CALCITE_MID1'
  INTERFACE COATING +BARE AIR CALCITE_EXT
  BOUNDS 0.2 -0.3

SURFACE
  PLANE Y -(5-(AIR_GAP/2))/TAN[38.5]
  PLANE Z 10
  PLANE Z 5+(AIR_GAP/2) RECT  5 25
  ROTATE X 38.5 0 5+(AIR_GAP/2)
OBJECT 'CALCITE_MID2'
  INTERFACE COATING +BARE AIR CALCITE_EXT
  BOUNDS -0.2 0.3

SURFACE
  PLANE Y -(5-(AIR_GAP/2))/TAN[38.5]
  PLANE Z 5+(AIR_GAP/2)
    ROTATE X 38.5 0 5+(AIR_GAP/2)
  PLANE Z 10  RECT  5 25
OBJECT 'CALCITE_BACK'
  INTERFACE COATING +BARE AIR CALCITE_EXT
  BOUNDS 0.2 0.3

SURFACE
  PLANE Z 0
  PLANE Z 5-(AIR_GAP/2)
    ROTATE X 38.5 0 5-(AIR_GAP/2)
  PLANE Y (5-(AIR_GAP/2))/TAN[38.5]  RECT  25 5
OBJECT 'CALCITE_TOP'

  INTERFACE COATING +BARE AIR CALCITE_EXT
  BOUNDS -0.2 0.3

SURFACE
  PLANE Z 10
  PLANE Z 5+(AIR_GAP/2)
    ROTATE X 38.5 0 5+(AIR_GAP/2)
  PLANE Y -(5-(AIR_GAP/2))/TAN[38.5] RECT  25 5  
OBJECT 'CALCITE_BOTTOM'
  INTERFACE COATING +BARE AIR CALCITE_EXT
  BOUNDS 0.2 -0.3

POLARIZ X 1 1 !!.37 .6`.2
RAYSET Z -5
0 0 1
SOURCE DIRECTION 0 0 1

WINDOW Y -2@1 X -2@1
PLOT POLAR  '+45 DEGREE POLARIZED BEAM BEFORE TRACE'
LIST ELLIPSE

WINDOW Y Z
TITLE '+45 DEGREE POLARIZED BEAM TRACED THROUGH A GLAN FOUCAULT PRISM'
PLOT FACETS OVERLAY  
TRACE PLOT

CONSIDER ONLY CALCITE_TOP
LIST ELLIPSE
WINDOW Z 3 4 X -2@.5
PLOT POLAR 'ORDINARY RAY'

CONSIDER ONLY CALCITE_BACK
LIST ELLIPSE
WINDOW Y -2@1 X -2@1
PLOT POLAR 'EXTRAORDINARY RAY'

Ideal Quarter-Wave Plate

Uses the IDEAL command to construct a quarter-wave plate, traces rays, and 
generates a polarization map of transmitted beam. Because the IDEAL command 
was used, there is no ray spitting and the PLOT POLAR command gives the same 
result for the transmitted polarization map as the FIELD... DELTA command.

!!Conversion of Linear to Circular Polarized Light with IDEAL Lens (ASAP Example)
SYSTEM NEW
RESET

UNITS MM
FRESNEL BOTH

WAVELENGTH 0.6328 UM
BEAMS COHERENT DIFFRACT
PARABASAL 4
WIDTHS 1.6

LENSES
  IDEAL Z 1 5
  1 0 0 1
  COS(45]`SIN(45] 0 0 SIN(45]`(-1*COS(45])
OBJECT 'QUARTER_WAVE_PLATE'

SURFACE
  PLANE Z 5 RECT 10
OBJECT 'DETECTOR'

POLARIZ X 1 1
GRID ELLIPTIC Z 0 -4@4.6 2@10
SOURCE DIRECTION 0 0 1

TITLE 'INPUT POLARIZATION STATE'
WINDOW Y X
PLOT POLAR

TRACE

SEGMENTS 2
ARROWS 1
TITLE 'POLARIZATION STATE AFTER QUARTER WAVE PLATE'
PLOT POLAR

PIXELS 15
TITLE 'ELECTRIC FIELD VECTOR'
FIELD REAL 5 DELTA

$DO 1 4
FIELD REAL 5 DELTA ?/8 OVERLAY  COLOR ?
RETURN

Wollaston Prism

Uses the CRYSTAL command to construct a Wollaston prism, ray traces, and shows 
a polarization map of the split rays.

!!ASAP FILE OF A WOLLASTON POLARIZING BEAM SPLITTER
SYSTEM NEW
RESET

TITLE 'BIREFRINGENT RAY SPLITTING IN A WOLLASTON PRISM'

MEDIA
  1.66
  1.49 CRYSTAL 1 0 0 1 'X_CALCITE'
  1.49 CRYSTAL 0 1 0 1 'Y_CALCITE'

BEAMS INCOHERENT GEOMETRIC
FRESNEL BOTH
SPLIT 2

SURFACE
  PLANE Z 0 RECT 1/TAN[15]
OBJECT 'FRONT'
  INTERFACE COAT +BARE AIR Y_CALCITE

SURFACE
  PLANE Z 1 RECT 1/SIN[15]; ROTATE X 15
OBJECT 'MIDDLE'
  INTERFACE COAT +BARE Y_CALCITE X_CALCITE

SURFACE
  PLANE Z 2 RECT 1/TAN[15]
OBJECT 'BACK'
  INTERFACE COAT +BARE X_CALCITE AIR

POLARIZ Y 1 1'90
RAYSET Z -1
0 0 1
SOURCE DIRECTION 0 0 1

WINDOW Y -2@1 X -2@1
PLOT POLAR 'INPUT POLARIZATION STATE'

WINDOW Y -1 1 Z -1 3

PROFILES OVERLAY
TRACE PLOT

CONSIDER ONLY 0.1

WINDOW Y X
WINDOW 1.5
PLOT POLARIZATION

Rhomb

Demonstrates the conversion of linear-to-circular polarization due to phase shifts 
on reflection from the surfaces of a Fresnel rhomb.

!!ASAP EXAMPLE OF FRESNEL RHOMB TO DEMONSRATE
!!POLARIZATION BY REFLECTION
SYSTEM NEW
RESET

BEAMS INCOHERENT GEOMETRIC
FRESNEL BOTH !!CONFIGURE POLARIZATION RAY TRACE
SPLIT 1

MEDIA
  1.517 'GLASS'

EDGES
  RECT Z 0 .5 .5 4
  REPEAT
  SHIFT Z 2
  SKEW Y 90-54.62 Z 0
OBJECT
  0.1 0.2 'PRISM_SIDES'
INTERFACE COATING BARE AIR GLASS

EDGES
  REPEAT 1
OBJECT 'FRONT_FACE'
  INTERFACE COATING BARE AIR GLASS

EDGES
  REPEAT
  SHIFT Z 2
 xiv ASAP Technical Guide



. .
 . 

. .
  SKEW Y 90-54.62 Z 0
OBJECT 'BACK_FACE'
INTERFACE COATING BARE AIR GLASS

EDGES
  REPEAT 0.1
  SHIFT Z 1
OBJECT 'EXIT_PLANE'

!!SET UP +45 DEGREE LINEARLY POLARIZED SOURCE
POLARIZ X 1 1
GRID POLAR Z -.5 0 .5 0 360 1 4 1
SOURCE DIRECTION 0 0 1

WINDOW Y X
WINDOW .5
ARROWS 3
SEGMENTS 3
PLOT POLAR 'INPUT 45 DEGREE LINEAR POLARIZATION'

WINDOW Y -0.5 2 Z -0.5 3
ARROWS 1
TITLE 'TOTAL INTERNAL REFLECTION FRESNEL RHOMB'
PLOT FACETS 1 1 OVERLAY
TRACE PLOT

SELECT ONLY 0 C !!SELECT ONLY THOSE RAYS IN +Z DIRECTION
WINDOW Y X
WINDOW 0.5
ARROWS 3
PLOT POLAR 'OUTPUT CIRCULAR POLARIZATION'
$VIEW

Polarizer Rotator

Uses the IDEAL command to construct a polarization rotator, and traces rays 
through rotator— reflecting from mirror, back through the rotator, and to a 
detector. It uses the LIST ELLIPSE command to examine the starting polarization 
and the polarization after each pass through the rotator. It shows that this type of 
rotator behaves like a Faraday rotator, in that two passes through the device double 
the effect rather than cancel it.

!! POLARIZATION ROTATOR
SYSTEM NEW

RESET

FRESNEL BOTH
SPLIT 2

ALPHA=180 !! OPTICAL ROTARY POWER IN DEGREES/MM
THICKNESS=.125 !! ROTATOR THICKNESS IN MM
THETA=ALPHA*THICKNESS

LENSES
  IDEAL Z 1 1
  1 0 0 1
  COS[THETA] SIN[THETA] -SIN[THETA] COS[THETA] !!FLIP
OBJECT 'ROTATOR'

SURFACE
  PLANE Z 2 RECT 1
OBJECT 'MIRROR'
  INTERFACE 1 0 AIR AIR

SURFACE
  PLANE Z -1 RECT 1.5
OBJECT 'DET'

POLARIZ X 1 0
GRID RECT Z .5 -4@1 1 2
SOURCE DIRECTION 0 0 1
!!SOURCE DIRECTION 0 SIN[5] COS[5]

LIST ELLIPSE

WINDOW Y Z
PLOT POLAR OVERLAY
PLOT FACETS OVERLAY
TRACE 0 MIRROR PLOT
PLOT POLAR
LIST ELLIPSE

WINDOW Y Z
TRACE PLOT
PLOT POLAR
$VIEW

LIST ELLIPSE

Calculating Stokes Parameters

We can use the IDEAL lens to compute the Stokes vector. Recall that the Stokes 
vector defines a vector space that spans polarized, unpolarized, and partially 
polarized light. The Stokes vector is determined by measuring four polarization 
parameters of the light. All four polarization parameters are determined by 
individually inserting a filter into the beam and measuring the transmitted 
irradiance. The four filters are:

1 Isotropic filter (passes all polarization states equally)

2 Horizontal linear polarizer

3 Linear polarizer at 45 degrees

4 Right hand circular polarizer

The Stokes vector elements are computed from the irradiances as follows:

5 S0 = IH + IV

6 S1 = IH - IV

7 S2 = I45  I135

8 S3 = IR - IC

The Stokes filters are inserted (and rotated if required) separately into a ray of 
polarized light with the aid of a $DO loop. The $DO loop also traces the rays once 
through each filter. All the filters exist in the object database, but are turned off and 
on with the CONSIDER command. The flux of each ray is grabbed after 
propagating through each filter, and the Stokes parameters and degree of 
polarization are computed with the above equations.

!!FILE FOR CALCULATING STOKES PARAMETERS
SYSTEM NEW
RESET

UNITS MM
FRESNEL BOTH

COATING
  0 1   'TRANS_50'

!!ISOTROPIC FILTER
LENS

  IDEAL Z 0 10
  1 0 0 1
OBJECT 'ISOTROPIC'
  INTERFACE COATING TRANS_50 AIR AIR
 
!!HORIZONTAL LINEAR POLARIZER
LENS
  IDEAL Z 1 10
  1 0 0 1
  1 0 0 0
OBJECT 'HOR_LIN_POL'
  INTERFACE COATING TRANS_50 AIR AIR

!!VERTICAL LINEAR POLARIZER
LENS
  IDEAL Z 2 10
  1 0 0 1
  0 0 0 1
OBJECT 'VER_LIN_POL'
  INTERFACE COATING TRANS_50 AIR AIR

!!+45 DEGREE LINEAR POLARIZER
LENS
  IDEAL Z 3 10
  1 0 0 1
  1 0 0 0
OBJECT '45D_LIN_POL'
  INTERFACE COATING TRANS_50 AIR AIR
  ROTATE Z 45 0 0

!!+135 DEGREE LINEAR POLARIZER
LENS
  IDEAL Z 4 10
  1 0 0 1
  1 0 0 0
OBJECT '135D_LIN_POL'
  INTERFACE COATING TRANS_50 AIR AIR
  ROTATE Z 135 0 0

!!RIGHT CIRCULAR POLARIZER
LENS
  IDEAL Z 5 10
  1 0 0 1
  1/2 0`1/2 0`-1/2 1/2
OBJECT 'RC_POL'
  INTERFACE COATING TRANS_50 AIR AIR

!!LEFT CIRCULAR POLARIZER
LENS

. .
  IDEAL Z 6 10
  1 0 0 1
  1/2 0`-1/2 0`1/2 1/2
OBJECT 'LC_POL'
  INTERFACE COATING TRANS_50 AIR AIR

SURFACE
 PLANE Z 7 RECT 10
OBJECT 'DETECTOR'

EMIT_SOURCE { 3
POLAR X #1 #2
EMITTING DISK Z 0 2@1 #3 10 10
 FLUX TOTAL 100
} 
ENTER POLARIZ X a  COMPONENT>
ENTER POLARIZ X a' COMPONENT>
ENTER TOTAL NUMBER OF RAYS>  

NRAYS=10000

$DO 2 7
{$EMIT_SOURCE 1 1 (NRAYS)  
!!$EMIT_SOURCE 1 0`1 (NRAYS)
!!INSERT THE ACTUAL SOURCE ABOVE
WINDOW X Z
CONSIDER ONLY 0 ? 8
!!PLOT FACETS OVERLAY
TRACE !!PLOT (NRAYS)/10
CONSIDER ONLY 8
STATS
$GRAB 'TOTAL' 0 2 PFLUX?
$REG PFLUX?
RETURN
 }

IH=PFLUX2
IV=PFLUX3
I45=PFLUX4
I135=PFLUX5
IRC=PFLUX6
ILC=PFLUX7

S0=IH+IV
S1=IH-IV
S2=I45-I135

S3=IRC-ILC
$REG S0 S1 S2 S3

RETURN

